<commit_before>#lang scheme/base

(provide strip-context)

(define (strip-context e)
  (cond
   [(syntax? e)
    (datum->syntax #f
                   (strip-context (syntax-e e))
                   e
                   e)]
   [(pair? e) (cons (strip-context (car e))
                    (strip-context (cdr e)))]
   [(vector? e) (list->vector
                 (map strip-context
                      (vector->list e)))]
   [(box? e) (box (strip-context (unbox e)))]
   [(prefab-struct-key e)
    => (lambda (k)
         (apply make-prefab-struct
                (strip-context (cdr (vector->list (struct->vector e))))))]
   [else e]))<commit_msg>syntax: fixed 'strip-context' on prefab structs   fixes bug in prefab struct literal in scheme/load module<commit_after>#lang scheme/base

(provide strip-context)

(define (strip-context e)
  (cond
   [(syntax? e)
    (datum->syntax #f
                   (strip-context (syntax-e e))
                   e
                   e)]
   [(pair? e) (cons (strip-context (car e))
                    (strip-context (cdr e)))]
   [(vector? e) (list->vector
                 (map strip-context
                      (vector->list e)))]
   [(box? e) (box (strip-context (unbox e)))]
   [(prefab-struct-key e)
    => (lambda (k)
         (apply make-prefab-struct
                k
                (strip-context (cdr (vector->list (struct->vector e))))))]
   [else e]))
{'fraction_non_alphanumeric': 0.15587188612099645, 'fraction_numerical': 0.0, 'mean_word_length': 1.7038461538461538, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 15, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '2953770', 'n_tokens_mistral': 464, 'n_tokens_neox': 431, 'n_words': 118}